
Abstract

We argue that declarative programs can be writ-
ten in precise natural language and back this
claim by using a complex optimisation prob-
lem. The problem specification is expressed in
natural language and automatically translated
into an executable answer set program, and an
answer set solver is then used to find (optimal)
solutions for natural language questions. Our
approach enables subject matter experts to ex-
press their knowledge in a natural and truly
declarative notation without the need to encode
this knowledge in a formal way.

1 Introduction

Declarative programming involves stating what is
to be computed, but not how it is to be computed;
it is a programming paradigm that expresses the
logic of computation without describing its con-
trol flow (Lloyd, 1994). Logic programming lan-
guages (Körner et al., 2022) and functional pro-
gramming languages (Hu et al., 2015) belong to
the declarative programming paradigm and result
in code that is characterised by a high level of
abstraction. One of the main benefits of declar-
ative programming languages is their ability to
describe problems with less code than imperative
programming languages. Furthermore, declarative
languages are known to be elaboration-tolerant,
precise, and easier to optimise than imperative
languages. In the context of logic programming,
the development of the stable model semantics
for logic programs (Gelfond and Lifschitz, 1988)
has led to answer set programming (Janhunen and
Nimelä, 2016), a powerful model-based language
for knowledge representation and non-monotonic
reasoning with industrial applications (Falkner
et al., 2018). Writing an answer set program in-
volves identifying objects and the relations between
them and formally encoding this information as

facts, rules and constraints. However, the resulting
formal notation may be difficult to understand by
subject matter experts who have detailed knowl-
edge about the application domain but do not have
a background in formal logic. Instead of encoding
a problem specification in answer set programming
notation, we suggested in previous work to express
such a specification in a precise subset of natural
language (Schwitter, 2018). Such a precise subset
of natural language is also known as a controlled
natural language (Kuhn, 2014). It has been shown
that the writing of a specification in controlled lan-
guage can be supported by a predictive authoring
tool like the writing of code with the help of a code
editor (Schwitter, 2020; Schwitter et al., 2003).
That means the authoring tool instructs the user
about the language constructs that can be used to
construct a textual specification. As in the case of
writing code, writing a specification in controlled
language is a process that requires careful planning
and an understanding of the application domain. In
this paper, we focus on an extension of our con-
trolled language PENGASP and illustrate how this
controlled language can be used to specify a com-
plex optimisation problem that goes beyond our
previous work but uses similar unification-based
techniques for the translation into an answer set
program (Guy and Schwitter, 2017). The novelty
is the use of choice rules, aggregates, and optimi-
sation statements that can be expressed directly on
the level of the controlled language. A state-of-the-
art answer set solver like clingo1 can then be used
to find the (optimal) solutions to the problem.

2 Answer Set Programming

Answer set programming (ASP) has its roots in the
fields of logic programming and non-monotonic
reasoning (Lifschitz, 2019; Gelfond and Kahl,

1https://potassco.org/clingo/

2014) and has been applied to a wide range of areas
in artificial intelligence, among them also to natu-
ral language processing tasks (Mitra et al., 2019;
Sharma, 2019; Schüller, 2018; Dzifcak et al., 2009).
ASP is supported by powerful reasoning tools and
offers a rich representation language that allows for
recursive definitions, (strong and weak) negation,
(strong and weak) constraints, aggregates, optimi-
sation statements, and external functions (Gebser
et al., 2019). An ASP program consists of a set of
rules of the following form:

l1 ;...; lm :-
lm+1,..., ln, not ln+1,..., not lo.

Here each li is a literal. A literal is either a pos-
itive atom of the form p(t1,...,tk) or its strong
negation -p(t1,...,tk), where p is a predicate
name and all ti are terms that are composed of func-
tion symbols and variables. The symbol :- (“if”)
separates the head of a rule from its body. The sym-
bol ; in the head of a rule stands for an (epistemic)
disjunction, and the symbol not in the body for
default negation (aka negation as failure). A rule
is called a fact if m = o = 1, normal if m = 1, and
an integrity constraint if m = 0. Semantically, the
rule above states that if lm+1,...,ln are true and
there is no reason to believe that ln+1,...,lo are
true, then at least one of l1,...,lm is believed to
be true.

To ease the use of ASP for practical applications,
several simplifying notations and extensions have
been developed (see (Gebser et al., 2019) for de-
tails). The most notable ones in our context are:
choice rules, aggregates, and objective functions.

A choice rule has the following form:

s { e1 ;...; em } t :- body.

Here ei is a choice element of the form
a:L1,...,Lk, where a is an atom; Li are possible
default-negated literals; and s and t are integers
which express lower and upper bounds on the cardi-
nality of elements. Intuitively, a choice rule means
that if the body of the rule is true, then an arbitrary
number of elements can be chosen as true as long
as this number complies with the lower and upper
bounds. Note that if ei ≥ 2 and s = t = 1, then
a choice rule implements an exclusive disjunction;
similarly, if s = 1, t = nil, then it implements
an inclusive disjunction. Aggregates are functions
that apply to sets and can be used to calculate for
example the number of elements of a set. For in-
stance, the expression:

#count { X, Y : p(X, Y) }

represents the number of elements of the set p/2.
Expressions like this can be used in the body of
an ASP rule as one side of a comparison, with a
variable on the other side, for example:

number of elements(N) :-

N = #count { X, Y : p(X, Y) }.

When an ASP program has several answer sets,
we may be interested in finding the best possible
one, according to some measure of quality. Objec-
tive functions can be used in this case to minimise
or maximise the sum of a set of weighted tuples (wi,
ti) that are subject to some conditions ci. These ob-
jective functions are expressed in ASP as directives
of the following form:

#minimize { w1@l1, t1 : c1 ;...;

wn@ln, tn : cn }.

Note that wi is a numerical constant, li is an op-
tional (lexicographically ordered) priority level, ti

is a sequence of terms, and ci is a sequence of
possibly default-negated literals. Alternatively, op-
timisation statements can be implemented as weak
constraints. In contrast to integrity constraints that
weed out answer sets as solutions, weak constraints
rank solutions.

3 Finding an Optimal Accommodation

Let us assume a traveller wants to choose the best
one among three different accommodations (Aloe,
Metro, Oase); each of them comes with a star rating
and a weekly room rent. Furthermore, one of the
accommodations is located on the main street and
known to be noisy. Considering the available op-
tions, the traveller faces the following optimisation
problem: (a) minimising noise has the highest prior-
ity; (b) minimising the cost per star has the second
highest priority; and (c) maximising the number
of stars of an accommodation that is otherwise not
distinguishable has the lowest priority.2

We can start expressing the factual information
about these three different accommodations in a
precise way in controlled language:

1. The bedroom apartment Oase is rated three
stars and costs 240 dollars.

2. The bedroom apartment Aloe is rated two
stars and costs 160 dollars.

2This example is inspired by (Gebser et al., 2019).



3. The studio apartment Metro that is located on
the main street is rated three stars and costs
200 dollars.

Furthermore, we specify ontological statements
that are necessary to solve the problem directly in
controlled language and describe what counts as an
accommodation (4 and 5); as a noisy accommoda-
tion (6); as the cost per star of an accommodation
(7); and as the star rating of an accommodation (8):

4. Every studio apartment is an accommodation.
5. Every bedroom apartment is an accommoda-

tion.
6. If an accommodation is located on a main

street then the accommodation is noisy.
7. If an accommodation costs N dollars and is

rated M stars then N / M is the cost per star
of the accommodation.

8. If an accommodation is rated N stars then N
is the start rating of the accommodation.

Next, we specify that one of the accommodations
is the optimal one, using an exclusive disjunction
in controlled language:

9. Either one of Aloe or Metro or Oase is opti-
mal.

The relevant optimisation statements are ex-
pressed with the help of predefined key phrases
Minimise/Maximise with a priority of I that include
a priority level, where a higher integer (I) indicates
a higher priority:

10. Minimise with a priority of 3 that an optimal
accommodation is noisy.

11. Minimise with a priority of 2 that C is the cost
per star of an optimal accommodation.

12. Maximise with a priority of 1 that S is the star
rating of an optimal accommodation.

Finally, the questions to be answered can be ex-
pressed as follows:

13. How many accommodations are there?
14. Which accommodation is optimal?

This entire specification can be translated auto-
matically into an executable ASP program. The
three factual statements (1-3) result in a number
of ASP facts. In our case, these facts are based on
a reified notation that relies on a small number of
predefined predicates. Constants that start with c

followed by a positive integer I are Skolem con-
stants and replace existentially quantified variables.

class(c1, bedroom_apartment). % 1
named(c1, oase).
prop(c1, c2, rated).
data_prop(c2, 3, cardinal).
class(c2, star).
pred(c1, c3, cost).
data_prop(c3, 240, cardinal).
class(c3, dollar).

class(c4, bedroom_apartment). % 2
named(c4, aloe).
prop(c4, c5, rated).
data_prop(c5, 2, cardinal).
class(c5, star).
pred(c4, c6, cost).
data_prop(c6, 160, cardinal).
class(c6, dollar).

class(c7, studio_apartment). % 3
named(c7, metro).
prop(c7, c8, located_on).
class(c8, main_street).
prop(c7, c9, rated).
data_prop(c9, 3, cardinal).
class(c9, star).
pred(c7, c10, cost).
data_prop(c10, 200, cardinal).
class(c10, dollar).

The ontological statements (4-8) result in five
ASP rules that define classes and properties:

class(A, accommodation) :- % 4
class(A, studio_apartment).

class(B, accommodation) :- % 5
class(B, bedroom_apartment).

prop(C, noisy) :- % 6
class(C, accommodation),
prop(C, D, located_on),
class(D, main_street).

prop(E/F, G, cost_per_star) :- % 7
class(G, accommodation),
pred(G, H, cost),
data_prop(H, E, cardinal),
class(H, dollar),
prop(G, I, rated),
data_prop(I, F, cardinal),
class(I, star).

prop(J, K, star_rating) :- % 8
class(K, accommodation),
prop(K, L, rated),
data_prop(L, J, cardinal),
class(L, star).

The two rules for the ontological statements 7
and 8 are interesting since both of them contain
an atom as rule head that has been derived from a
relational noun (cost per star and star rating) and
introduce properties. The property that represents
cost per star contains an arithmetic function (E/F)
with two variables as its first argument. This arith-
metic function picks up two cardinal numbers and
evaluates their ratio during grounding. The prop-



erty that represents star rating picks up a cardinal
number (J) as its first argument.

The statement 9 is translated into a choice rule.
The integers before and after the expression in the
braces express lower and upper bounds on the car-
dinality. In our case, the lower bound and upper
bound is 1, meaning that exactly one accommoda-
tion is optimal:

1 { prop(M, optimal) : % 9
named(M, (aloe ;

metro ;
oase)) } 1.

The optimisation statements (10-12) are trans-
lated into weak constraints with the help of
#minimize and #maximize directives. The first ar-
gument w@l of these directives consists of a weight
(w) and priority level (l), greater levels being more
significant than smaller ones. These directives in-
struct clingo to look for the best stable model of
the given ASP program.

#minimize { 1@3, % 10
N : prop(N, optimal),

class(N, accommodation),
prop(N, noisy) }.

#minimize { O@2, % 11
P : prop(O, P, cost_per_star),

prop(P, optimal),
class(P, accommodation) }.

#maximize { Q@1, % 12
R : prop(Q, R, star_rating),

prop(R, optimal),
class(R, accommodation) }.

Questions such as (13 and 14) are translated
into an ASP rule with a specific answer literal
(answer/1) as head. These literals will contain
the answer to the question after grounding and will
be displayed using the #show directive (15):

answer(T) :- T = #count { % 13
S : class(S, accommodation) }.

answer(V) :- % 14
named(U, V),
class(U, accommodation),
prop(U, optimal).

#show answer/1. % 15

Note that question (13) is not necessary to find
the optimal solution but illustrates the use of an
aggregate construct.

4 Evaluation

If we submit our ASP program to the answer set
solver clingo, then the solver will generate and
display three answer sets (models), one for each

accommodation together with the weights used for
finding the optimal solution. These answer sets
also contain the answers to the questions (13 and
14) that are displayed with the help of the #show

directive (15):
clingo version 5.6.1
Reading from asp.lp
Solving...
Answer: 1
answer(3) answer(metro)
Optimization: 1 66 -3
Answer: 2
answer(3) answer(aloe)
Optimization: 0 80 -2
Answer: 3
answer(3) answer(oase)
Optimization: 0 80 -3
OPTIMUM FOUND

Models : 3
Optimum : yes

Optimization : 0 80 -3
Calls : 1
Time : 0.037s (...)
CPU Time : 0.000s

We can see in the output that the accommodation
Metro is noisy (1) and therefore not optimal with
respect to the most important optimisation state-
ment. The accommodations Aloe and Oase are not
noisy (0) and are both optimal with respect to the
cost per star ratio (80); the second most important
optimisation statement. This tie is broken by the
least important optimisation statement that looks at
the number of stars. Note that since we maximise
the number of stars, the values are displayed as neg-
ative integers (-2 and -3). In summary, the optimal
accommodation is Oase, since it is not noisy, has a
cost per star ratio of 80 and is rated three stars.

5 Conclusion

We showed in this paper that complex optimisa-
tion statements can be expressed directly in precise
natural language. The resulting specification can
then be automatically translated into an executable
ASP program. The writing of such a specification
in controlled language is usually supported by a
predictive authoring tool that has similar features
as a code editor for a programming language. As
in the case of writing declarative code, writing a
textual specification in controlled language needs
to be carefully planned and will never be a fast pro-
cess, but our approach has the potential to close the
gap between a (seemingly) informal textual spec-
ification and a declarative program. In this sense,
programming in controlled language is the most
extreme form of declarative programming.

